Fréchet distance approximation and Maple applications

Mira-Cristiana Anisiu and Valeriu Anisiu
Tiberiu Popoviciu Institute of Numerical Analysis
Faculty of Mathematics and Computer Science
Babeş-Bolyai University

GDO 2019
GAMES, DYNAMICS AND OPTIMIZATION
APRIL 9-11, 2019

1 Introduction

Maurice Fréchet defined the distance (named in French *écart*) which now bears his name for curves in $\mathbb{R}^3$ and proved that it satisfied the axioms of the distance.

Maurice René Fréchet was born September 10, 1878, at Maligny. While being a student to *Lycée Buffon* in Paris he was taught Mathematics in 1890-1893 by Jacques Hadamard, who tutored him individually. Hadamard was appointed to a professorship in the *University of Bordeaux* from 1894 to 1897, and then in Paris. They continuously wrote to each other on mathematical problems, or met in Paris for discussions. Fréchet entered the *École Normale Supérieure* in 1900 after doing his military service. He completed his course there in 1903, and he was the first of Hadamard doctoral students, obtaining his Ph.D in 1906 at the University of Bordeaux.

Fréchet intended to give definitions in general spaces, and he did this in [6], where he consider the notion of *écart*. To two elements A and B (which can be points, curves, functions) there corresponds a number $(A, B) \geq 0$, so that:

- $(A, B) = 0$ if A and B are not distinct and only in this case;

- A, B, C being three arbitrary elements, if the écarts (A, C) and (B, C) are infinitely small, the same is true for (A, B). In this short note he gave a sketch of the definition of the distance between two curves in $\mathbb{R}^3$.

He then gave the definition of the distance for curves in $\mathbb{R}^3$ in the longer paper [7] published in America, where he proved its properties. During his studies at the *École Normale Supérieure*, Fréchet get acquainted with E. B. Wilson, who become an editor of the *Transactions of the American Mathematical Society* from 1903 to 1919. He published several papers in *Transactions* and this helped to make his work known to American mathematicians. In a third paper dedicated to continuous curves [8], Fréchet studied some topological properties. These works were included in his doctoral Thesis submitted on April 2 1906. In his dissertation, Fréchet introduces the definition of the metric space, whose name was given by Hausdorff - *metrischer Raum* ([10]) and also that of completeness. The class of the continuous curves in $\mathbb{R}^3$ serves as an example for his abstract theory.

The life and work of Maurice Fréchet from this period are presented in the first essay [12] out of three dedicated to Fréchet's role in the history of functional analysis by the American professor A.E. Taylor. A Colombian student, L.C. Arboleda, put Fréchet's papers from the Archives of the Académie des Sciences in good order, while working for his doctoral thesis defended in 1980, *Contributions à l'Étude des Premières Recherches Topologiques (d'après la correspondence et les publications de Maurice Fréchet, 1904-1928)*. A description of the files from the archive is to be found in [4].

Fréchet was a school teacher at Besançon and Nantes, and then professor at the Universities of Poitiers, Strasbourg and Paris. He died June 4, 1973, in Paris.

2 The Fréchet distance

Let (X, d) be a metric space. A curve A in X is a continuous map from the unit interval into X, i.e., $A : [0, 1] \to X$. A reparametrization α of the unit interval is a continuous, non-decreasing surjection $\alpha : [0, 1] \to [0, 1]$. For two given curves A and B in X, the *Fréchet distance* $F(A, B)$ between them is defined as the infimum over all reparametrizations α and β of $[0, 1]$ of the maximum over all $t \in [0, 1]$ of the distance in X between $A(\alpha(t))$ and $B(\beta(t))$,

$$F(A, B) = \inf_{\alpha,\beta} \max_{t\in[0,1]} \{d(A(\alpha(t)), B(\beta(t)))\} .$$

It is a good measure for the resemblance of two curves, from this point of view better then the Hausdorff distance, because it takes into account the flow of the two curves. For two curves $A, B : [0,1] \to X$, the *Hausdorff distance* (or *Pompeiu-Hausdorff distance*) is given by

$$H(A,B) = \max\{\sup_{x\in[A]} \inf_{y\in[B]} d(x,y), \sup_{y\in[B]} \inf_{x\in[A]} d(x,y)\}.$$

Here, $[A]$ and $[B]$ are the supports of the curves. This distance was defined in [11] by Pompeiu for sets in the complex plane, and in [10] by Hausdorff in metric spaces.

It is obvious that

$$H(A,B) \leq F(A,B).$$

In [1] the equality between the Fréchet and Hausdorff distances for the boundary of a convex set in the plane is proved. The parametrizations of the boundary ∂C of a convex set C are supposed to be injective. The following theorem holds.

Theorem 1 *Let C_1 and C_2 be convex sets in the plane. Then $F(\partial C_1, \partial C_2) = H(\partial C_1, \partial C_2) = H(C_1, C_2)$, where F is taken with respect to any two monotone parametrizations.*

A similar result was proved later for closed convex curves in [2]. A *simple closed curve* is a subset of $\mathbb{R}^2$ that is homeomorphic to the circle S^1, and assume that simple closed curves are parametrized by such a homeomorphism. Assume also that a fixed orientation of the parameter space S^1 is given and that the homeomorphism is orientation preserving. A simple closed curve is called a *convex closed curve* if it encloses a convex set. Therefore the boundary ∂C of any bounded convex set $C \subset \mathbb{R}^2$ is a closed convex curve.

In the definition of the Fréchet distance of two simple closed curves, the parametrizations α and β are supposed to be orientation preserving homeomorphisms on S^1. The following theorem holds.

Theorem 2 *For any closed convex curves A and B, $F(A, B) = H(A, B)$.*

Remark 3 *Monotone (non-strict) parametrizations have been considered by Fréchet himself. He wanted to include in the approach curves $A : [0, 1] \to X$ which are constant in an interval. Note that there can be even countably many such intervals. A classical example is Cantor's function (*Devil's staircase*) which, by the way, can be defined and represented by a very short Maple procedure.*

```
> Cf:=proc(x, n:=floor(Digits/log10(2)+2))
  if not(type(x,realcons)) then return 'Cf'(x) fi;
  if x=0 or x=1 then return x fi; if n<0 then return 0.0 fi;
  piecewise(x<1/3, Cf(3*x,n-1)/2, x<=2/3, 1/2, 1/2+Cf(3*x-2,n-1)/2)
  end:
```

```
> plot(Cf,0..1);
```




```
> Cf(80/81);
```

$$15/16$$

```
> Cf(1/2);
```

$$1/2$$

```
> evalf(Int(Cf, 0..1));
```

$$0.5000000000$$

```
> evalf(Int(Cf^2, 0..1, epsilon=1e-4));
```

$$0.2999991429$$

3 Approximations of the Fréchet distance

However, the Fréchet distance is difficult to compute.

To find $F(A, B)$ one considers polygonal approximations of the curves A and B, as suggested in [5], and try to find algorithms which allow the approximation of the Fréchet distance. Using the Computer Algebra System Maple (see for example [3]) we can calculate the Fréchet distance between several curves by implementing rather fast methods.

3.1 Free-space diagram

The **free-space diagram**, introduced by Alt and Godau is the set

$$FS(A, B, \varepsilon) = \{(a, b) \in [0, 1]^2 : d(A(a), B(b)) \leq \varepsilon\},$$

where A, B are parametrizations (in the unit interval) for two curves in a metric space (X, d).

The Fréchet distance is at most ε if and only if $FS(A, B, \varepsilon)$ contains a path from the lower left corner to the upper right corner, which is monotone both in the horizontal and in the vertical direction.

```
> A:=x->x*(sin(14*x)+2)^2/9+1/10;   # dF=0.14...
> B:=x->x*(sin(15*x)*4/3+2+sin(20*x)/10)^2/9
```

$$A := x \to \frac{1}{9}\, x \left(\sin\left(14\, x\right) + 2\right)^2 + \frac{1}{10}$$

$$B := x \to \frac{1}{9}\, x \left(\frac{4}{3}\, \sin\left(15\, x\right) + 2 + \frac{1}{10}\, \sin\left(20\, x\right)\right)^2$$

```
> plot([A,B], 0..1, color=[red,blue]);
```



```
FreeSpace :=proc(A,B, epsilon, N:=400)
local i,j, e2:=epsilon^2,
      X:=Vector(),Y:=Vector(), n:=0, yi,yj;
for i from 0 to N do
for j from 0 to N do
  yi:=evalf(A(i/N)); yj:=evalf(B(j/N));
  if (yi-yj)^2 + ((i-j)/N)^2 <= e2 then
    n:=n+1;  X(n):=i/N; Y(n):=j/N;
  fi;
od od;
print('n'=n);
plots:-display(plot([[0,0],[1,0],[1,1],[0,1],[0,0]],color=black),
               plots:-pointplot(X,Y,color=red),gridlines);
end:
```

```
> FreeSpace(A,B, 0.15);
```



```
> FreeSpace(A,B, 0.14);
```



```
> plot(abs@(A-B), 0..1);
```



3.2 Discrete Fréchet distance

In computing the Fréchet distance between arbitrary curves one typically approximates the curves by polygonal curves. A *polygonal curve* in a metric vector space (X, d) is a curve $P : [1, n] \to X$, where n is a positive integer, such that for each $i \in \{1, ..., n-1\}$, the restriction of P to the interval $[i, i+1]$ is affine, i.e. $P(i + \lambda) = (1 - \lambda)P(i) + \lambda P(i + 1)$.

Let $P : [1, n] \to V$ be a polygonal curve. We denote the sequence $(P(1), ..., P(n))$ of vertices (endpoints of the line segments of P) by $V(P)$. Also, $d(P)$ denotes $\max_{1\leq i\leq n-1} d(P(i), P(i+1))$.

Let P and Q be polygonal curves and $V(P) = (u_1, ..., u_p)$ and $V(Q) = (v_1, ..., v_q)$ the corresponding sequences. A *coupling* L between P and Q is a sequence

$$(u_{a_1}, v_{b_1}), (u_{a_2}, v_{b_2}), ..., (u_{a_m}, v_{b_m})$$

of distinct pairs from $V(P) \times V(Q)$ such that $a_1 = 1$, $b_1 = 1$, $a_m = p$, $b_m = q$, and for all $i = 1, ..., q$ we have $a_{i+1} = a_i$ or $a_{i+1} = a_i + 1$,and $b_{i+1} = b_i$ or $b_{i+1} = b_i + 1$. Thus, a coupling has to respect the order of the points in P and Q. The length $||L||$ of the coupling L is the length of the longest link in L, that is,

$$||L|| = \max_{i=1,...,m} d(u_{a_i}, v_{b_i}).$$

Given the polygonal curves P and Q, their *discrete Fréchet distance* F^* is defined to be

$$F^*(P, Q) = \min\{||L|| : L \text{ is a coupling between } P \text{ and } Q\}.$$

```
dF:=proc(P::listlist, Q::listlist)
    # input: polygons  P,Q;
    # return: F*(P, Q) = the discrete Frechet distance
    #         and the optimal coupling (as two lists)
    local A := Matrix(nops(P),nops(Q),-1), c, df,
          iP:=Vector(nops(P)+nops(Q)), jQ:=copy(iP), m:=1, i:=nops(P),j:=nops(Q);

    c:=proc(i, j)
       if A[i,j] > -1 then  return A[i,j];
       elif i = 1 and j = 1 then A[i,j] := d(P[1], Q[1])
       elif i > 1 and j = 1 then A[i,j] := max( c(i-1,1), d(P[i], Q[1]) )
       elif i = 1 and j > 1 then A[i,j] := max( c(1,j-1), d(P[1], Q[j]) )
       elif i > 1 and j > 1 then A[i,j] :=
             max( min(c(i-1,j), c(i-1,j-1), c(i,j-1)), d(P[i], Q[j]) )
       else A[i,j] := infinity;  print([i,j]=infinity);
       fi;
    return A[i,j];
    end; # function c


  df := c(nops(P), nops(Q));
  iP[m]:=nops(P); jQ[m]:=nops(Q);
```

```
  while (i>1) or (j>1) do
     m:=m+1;
     if   i=1 then j:=j-1
     elif j=1 then i:=i-1
     elif A[i-1,j-1]<=df then i:=i-1; j:=j-1
     elif A[i,j-1]<=df then j:=j-1
     elif A[i-1,j]<=df then i:=i-1  fi;
  iP[m]:=i; jQ[m]:=j
  od;
  df, iP[1..m], jQ[1..m];
end:
```

To use this program, the distance d of the metric space must be defined. For example, for the Euclidean metric in $\mathbb{R}^2$, use

```
d := (u,v) -> sqrt( (u[1]-v[1])^2 + (u[2]-v[2])^2 );
```

The following short program produces an animation for two concentric circles with opposite senses. The Fréchet distance in this case is $R + r$. Note that if the senses are the same, the Fréchet distance and the Hausdorff-Pompeiu distances are both $R - r$.

```
cP:=t -> [cos(2*Pi*t), sin(2*Pi*t)];   # dF=0.14...
cQ:=t -> [2*cos(2*Pi*t), -2*sin(2*Pi*t)];
```

```
p:=400; q:=400;

P:=evalf([ seq(cP(i/p),i=0..p) ]):
Q:=evalf([ seq(cQ(i/q),i=0..q) ]):

df,a,b := dF(P,Q);
m:=numelems(a);

imax:=select( i -> d(P[a[i]],Q[b[i]])=df, [seq(1..m)]);
lnkmax:=seq( [ P[a[i]],Q[b[i]] ], i=imax);

plotPQ:=plot([P,Q], color=[red,blue], size=[1200,400]);
plotPQ1:=display(plotPQ,plot([lnkmax], thickness=4, color=gold));

animate( plot, [ [ 'P'['a'[round(i)]],'Q'['b'[round(i)]]], color=green, thickness=6 ],
         i=1..m, frames=100, background=plotPQ1, size=[800,800] );
```

The following results are from [5].

Theorem 4 *For any polygonal curves P and Q*

$$F(P,Q) \leq F^*(P,Q) \leq F(P,Q) + \max\{d(P), d(Q)\}.$$

Proposition 5 *Let $P_0, P_1, \ldots$ and $Q_0, Q_1, \ldots$ be two sequences of polygonal curves such that P_{i+1} (respectively Q_{i+1}) is a refinement of P_i (respectively Q_i) for all $i \geq 0$ and $\lim_{i\to\infty} d(P_i) = \lim_{i\to\infty} d(Q_i) = 0$. Then,*

$$\lim_{i\to\infty} F^*(P_i, Q_i) = F(P_0, Q_0).$$

Theorem 6 *Let $P : [1, p] \to X$ and $Q : [1, q] \to X$ be polygonal curves. The measure $F^*(P, Q)$ can be computed in $O(pq)$ time.*

The motivation for studying such measures comes from the notion of truthlikeness in philosophy of science and from artificial intelligence, where the task of quantifying the distance of theories is important for example for machine learning and theory approximation. Theories viewed as sets of models naturally correspond to curves viewed as sets of points. Given a metric on the models, the distance between the sets of models may be interpreted as a measure of similarity between theories.

References

[1] H. Alt, J. Blömer, H. Wagener, Approximation of convex polygons, In International Colloquium on Automata, Languages, and Programming 1990 Jul 16, 703-716, Springer, Berlin, Heidelberg

[2] H. Alt, C. Knauer, C. Wenk, Comparison of distance measures for planar curves, Algorithmica, 38 (1) (2004), 45-58

[3] V. Anisiu, *Calcul simbolic cu Maple*, Presa Universitară Clujeană, 2006

[4] L.C. Arboleda, Rapport sur l'inventaire et l'analyse des papiers du Fonds-Fréchet dans les archives de l'Académie des Sciences de Paris, Cahiers du séminaire d'histoire des mathématiques, 2 (1981), 9-17

[5] Th. Eiter, H. Manilla, Computing discrete Fréchet distance, Tech. Report CD-TR 94/64, Information Systems Department, Technical University of Vienna, 1994

[6] M. Fréchet, La notion d'écart dans le Calcul fonctionnel, Comptes Rendus Acad. Sci. 140 (1905), 772-774

[7] M. Fréchet, Sur l'écart de deux courbes et sur les courbes limite, Transactions of the American Mathematical Society 6 (4) (1905), 435-449

[8] M. Fréchet, Les ensembles de courbes continues, Comptes Rendus Acad. Sci. 141 (1905), 873-875

[9] M. Fréchet, Sur quelques points du calcul fonctionnel, Rendiconti del Circolo Matematico di Palermo 22 (1) (1906), 1-72

[10] F. Hausdorff, Grundzüge der Mengenlehre, Leipzig 1914

[11] D. Pompeiu, Sur la continuité des fonctions de variables complexes (Thèse), Gauthier-Villars, Paris, 1905; Ann. Fac. Sci. de Toulouse 7 (1905), 264-315

[12] A.E. Taylor, A study of Maurice Fréchet: I. His early work on point set theory and the theory of functionals. Archive for History of Exact Sciences, 27 (3) (1982), 233-295.